NonlinearModelFit Examples

- Constraining and Initializing Fit Parameters -

-Fitting Only Part of your data -

Bill Knowlton

Boise State University

January 18, 2015

August 23, 2017

November 10, 2017

Using NonlinearModelFit to fit data. Note that the data is linear just to make it easy to create, but it does not have to be. Also, rather than use the Show command, I created a table of fit data using the fit equation and plotted it as a line with the data using the ListPlot command.

In[997]:=
```
(*data and defining part of the data that is to be fit*)
Clear[data1, fitdata, fitdatatable]
data1 = {{1, 1}, {2, 2}, {3, 3}, {4, 4}, {5, 5}, {6, 6}, {7, 7}};

(*fitting data*)
fitdata = NonlinearModelFit[data1, a x + b, {{a, 2}, {b, 7}}, x];
(* Nonlinear fit of the data *)
fitdatatable = Table[{x, fitdata[x]}, {x, .5, 7.5, 0.1}];
(*Creating a table of data using the fit equation so that I can
 use it in ListPlot then join the data with a line to plot the fit *)

listplot = ListPlot[{data1, fitdatatable},
  PlotRange → {{0, 10}, {0, 10}}, Frame → True, GridLines → Automatic,
  PlotStyle → {Red, Blue, Blue}, Joined → {False, True}, PlotMarkers → {"●", ""},
  FrameLabel → {Style["X-axis (a.u.)", Medium], Style["Y-axis (a.u.)", Medium]},
  PlotLegends → Placed[PointLegend[{"Data", "Fit"}, Background → White,
     LegendFunction → "Frame", LegendLabel → "Legend"], {0.25, 0.8}], Epilog →
   {Green, PointSize[Large], Point[{4, 4}], Green, Text["Approx. Mid-Point", {2, 4}]}]

Print["Fit = ", Framed[fitdata["BestFit"]]] (*generates the fit equation *)
Print["R² = ", Framed[fitdata["RSquared"]]]
(*generates and prints the R^2 for the fit *)
Print["Parameter Table:"]
Print[fitdata["ParameterTable"]](*generates and prints a table of fit parameters *)
fitdata["BestFitParameters"] (* generates a list of fit parameters so
 that the values can be extracted to use in any following calculations *)
```

Out[1000]=



Fit = $-6.714\times10^{-16}+1.\,x$

R^2 = 1.

Parameter Table:

	Estimate	Standard Error	t-Statistic	P-Value
a	1.	1.85776×10^{-16}	5.38283×10^{15}	4.20001×10^{-78}
b	-6.714×10^{-16}	8.30815×10^{-16}	−0.808122	0.455733

Out[1005]= $\{a \to 1.,\ b \to -6.714\times10^{-16}\}$

Same approach, but used the Show command, which I don't think is quite as straightforward.

```
(*data and defining part of the data that is to be fit*)
data1 = {{1, 1}, {2, 2}, {3, 3}, {4, 4}, {5, 5}, {6, 6}, {7, 7}};

(*fitting data*)
fitdata = NonlinearModelFit[data1, a x + b, {{a, 2}, {b, 7}}, x];
(* Nonlinear fit of the data *)

(*plotting data and fits*)
listplot1 =
  ListPlot[data1, PlotRange → {{0, 10}, {0, 10}}, Frame → True, GridLines → Automatic,
   FrameLabel → {Style["x (a.u.)", Large], Style["y (a.u.)", Large]},
   PlotStyle → {PointSize[.05], Green}, PlotLegends → Placed[
     PointLegend[{"data"}, Background → White], {0.25, 0.85}]]; (* plots the data *)
plotfit = Plot[fitdata[x], {x, 0.5, 7.5}, PlotRange → {{0, 10}, {0, 10}},
   GridLines → Automatic, FrameLabel →
    {Style["x (a.u.)", Large], Style["y (a.u.)", Large]}, PlotStyle → {Cyan}, PlotLegends →
    Placed[LineLegend[{"fit"}, Background → White], {0.25, 0.75}]];(* plots the fit *)

Show[listplot1, plotfit](*Show command plots both the data and the fit together*)

Print["Fit = ", Framed[fitdata["BestFit"]]] (*generates the fit equation *)
Print["R^2 = ", Framed[fitdata["RSquared"]]]
(*generates and prints the R^2 for the fit *)
Print["Parameter Table:"]
Print[fitdata["ParameterTable"]](*generates and prints a table of fit parameters *)
fitdata["BestFitParameters"] (* generates a list of fit parameters so
 that the values can be extracted to use in any following calculations *)
```



Fit = $-6.714\times10^{-16}+1.\,x$

R^2 = 1.

Parameter Table:

	Estimate	Standard Error	t-Statistic	P-Value
a	1.	1.85776×10^{-16}	5.38283×10^{15}	4.20001×10^{-78}
b	-6.714×10^{-16}	8.30815×10^{-16}	−0.808122	0.455733

$\{a\to 1.,\ b\to -6.714\times10^{-16}\}$

NonlinearModelFit with Initial Values for Fit Parameters. Most of this example was taken from the Documentation Center Examples

```
data = {{25., 0.001}, {25.5, 0.002}, {26., 0.011}, {26.5, 0.045}, {27., 0.112}, {27.5, 0.215},
   {28., 0.259}, {28.5, 0.206}, {29., 0.112}, {29.5, 0.044}, {30., 0.011}};

(*fitting data*)
nlm = NonlinearModelFit[data, {a Exp[-(x - b)^2], a > 0, 30 > b > 20}, {{a, .5}, {b, 25}}, x]
(*Providing constraints and initial values for fitting parameters a & b*)
(*Some other variations:
   Without constraining and without initializing: nlm=
 NonlinearModelFit[data,a*Exp[-(x-b)^2],{a,b},x]
   With initializing but without constraining: nlm=
  NonlinearModelFit[data,a*Exp[-(x-b)^2],{{a,.5},{b,25}},x]
    Initializing and Constraining: nlm=
   NonlinearModelFit[data,{a*Exp[-(x-b)^2],a>0,30>b>20},{{a,.5},{b,25}},x] *)

(*Plotting data*)
Plot[nlm[x], {x, 25, 31}, Epilog :> Point[data], PlotStyle -> {Red, Thick}];
(*did not use, but nice way to ad data points to data*)
plotdata =
 ListPlot[data, PlotRange -> {{25, 31}, {0, 0.3}}, Frame -> True, GridLines -> Automatic,
  FrameLabel -> {Style["Wavelength (nm)", Large], Style["Intensity (a.u.)", Large]},
  PlotStyle -> {PointSize[0.03], Green}, PlotLegends ->
   Placed[PointLegend[{"X-ray data"}, Background -> White], {0.85, 0.85}]] (*data plot*)
plotfit = Plot[nlm[x], {x, 25, 31}, Frame -> True,
  PlotRange -> {{25, 31}, {0, 0.3}}, GridLines -> Automatic,
  FrameLabel -> {Style["Wavelength (nm)", Large], Style["Intensity (a.u.)", Large]},
  PlotStyle -> {Green}, PlotLegends ->
   Placed[LineLegend[{"X-ray fit"}, Background -> White], {0.85, 0.75}]] (*fit plot*)
Show[plotdata, plotfit, PlotLabel -> Style["Gaussian Fits of Data", Large]]
(*both plots together*)

(*Statistics of Fit*)
Print["Statistics of Fit"]
Print["R^2 = ", nlm["RSquared"]]
Print["Adjusted R^2 = ", nlm["AdjustedRSquared"]]
nlm["ParameterTable"]
nlm["ParameterConfidenceIntervalTable"]
nlm["BestFitParameters"]
nlm["BestFit"]
```

FittedModel[$0.270927\, e^{-(-27.988+x)^2}$]





Gaussian Fits of Data




```
Statistics of Fit
```

R^2 = 0.994334

Adjusted R^2 = 0.993075

FittedModel: The property values {ParameterTable} assume an unconstrained model. The results for these properties may not be valid, particularly if the fitted parameters are near a constraint boundary.

	Estimate	Standard Error	t-Statistic	P-Value
a	0.270927	0.00681714	39.742	2.01152×10^{-11}
b	27.988	0.0251627	1112.28	1.95402×10^{-24}

FittedModel: The property values {ParameterConfidenceIntervalTable} assume an unconstrained model. The results for these properties may not be valid, particularly if the fitted parameters are near a constraint boundary.

	Estimate	Standard Error	Confidence Interval
a	0.270927	0.00681714	{0.255505, 0.286348}
b	27.988	0.0251627	{27.931, 28.0449}

$\{a \to 0.270927, b \to 27.988\}$

$0.270927\, e^{-(-27.988+x)^2}$

Fitting only part of your data:

If one wants to fit only part of a data set, this example shows how to extract part of the data and fit it.

Commands used: NonlinearModelFit[ ] & Take[ ]

```
(*data and defining part of the data that is to be fit*)
Clear[data1, fitdatatable, PartOfData, fitpartdata, listplot]
data1 = {{0, 0}, {1, 1}, {2, 2}, {3, 3},
   {4, 9}, {5, 12}, {6, 15}, {7, 19}, {8, 29}, {9, 30}, {10, 31}};
PartOfData = Take[data1, {4, 8}] (*Takes part of the data so I can fit it independently
 of the entire data set. In this case, it is data point 4 thru data point 8*)

(*fitting data*)
fitpartdata = NonlinearModelFit[PartOfData, a x + b, {{a, 2}, {b, 7}}, x];
fitdatatable = Table[{x, fitpartdata[x]}, {x, 4, 7, 0.1}];

(*plotting data and fits*)
listplot1 = ListPlot[data1, PlotRange → {{0, 10}, {0, 32}}, Frame → True,
  GridLines → Automatic, FrameLabel → {Style["x (a.u.)", Large], Style["y (a.u.)", Large]},
  PlotStyle → {PointSize[0.03], Green},
  PlotLegends → Placed[PointLegend[{"data"}, Background → White], {0.25, 0.85}]]

listplot = ListPlot[{data1, fitdatatable},
  PlotRange → {{0, 12}, {-1, 35}}, Frame → True, GridLines → Automatic,
  PlotStyle → {Red, Blue, Blue}, Joined → {False, True}, PlotMarkers → {"●", ""},
  FrameLabel → {Style["X-axis (a.u.)", Medium], Style["Y-axis (a.u.)", Medium]},
  PlotLegends → Placed[PointLegend[{"Data", "Fit"}, Background → White,
     LegendFunction → "Frame", LegendLabel → "Legend"], {0.25, 0.8}],
  Epilog → {Green, PointSize[Large], Point[{5.6, 14}], Blue,
    Text["Fit Region", {7, 12}], Green, Text["Mid-Point", {4.5, 14}]}]


Print["Fit = ", fitpartdata["BestFit"]]
Print["R² = ", fitpartdata["RSquared"]]
Print["Parameter Table:"]
Print[fitpartdata["ParameterTable"]]
fitpartdata["BestFitParameters"]
```

Out[923]= {{3, 3}, {4, 9}, {5, 12}, {6, 15}, {7, 19}}

Out[926]=



Out[927]=



Fit = −7.4 + 3.8 x

R^2 = 0.996585

Parameter Table:

	Estimate	Standard Error	t-Statistic	P-Value
a	3.8	0.305505	12.4384	0.00111985
b	−7.4	1.58745	−4.66156	0.0186307

Out[932]= {a → 3.8, b → −7.4}

```
(*data and defining part of the data that is to be fit*)
data1 = {{0, 0}, {1, 1}, {2, 2}, {3, 3},
    {4, 9}, {5, 12}, {6, 15}, {7, 19}, {8, 29}, {9, 30}, {10, 31}};
PartOfData = Take[data1, {4, 8}] (*Takes part of the data so I can fit it independently
 of the entire data set. In this case, it is data point 4 thru data point 8*)

(*fitting data*)
fitpartdata = NonlinearModelFit[PartOfData, a x + b, {{a, 2}, {b, 7}}, x];

(*plotting data and fits*)
listplot1 = ListPlot[data1, PlotRange → {{0, 10}, {0, 32}}, Frame → True,
  GridLines → Automatic, FrameLabel → {Style["x (a.u.)", Large], Style["y (a.u.)", Large]},
  PlotStyle → {PointSize[0.03], Green},
  PlotLegends → Placed[PointLegend[{"data"}, Background → White], {0.25, 0.85}]]
plotfit = Plot[fitpartdata[x], {x, 4, 7},
   PlotRange → {{0, 10}, {0, 32}}, GridLines → Automatic,
   FrameLabel → {Style["x (a.u.)", Large], Style["y (a.u.)", Large]}, PlotStyle → {Green},
   PlotLegends → Placed[LineLegend[{"fit"}, Background → White], {0.25, 0.75}]];
Show[listplot1, plotfit]

Print["Fit = ", fitpartdata["BestFit"]]
Print["R^2 = ", fitpartdata["RSquared"]]
Print["Parameter Table:"]
Print[fitpartdata["ParameterTable"]]
fitpartdata["BestFitParameters"]
```

{{3, 3}, {4, 9}, {5, 12}, {6, 15}, {7, 19}}






```
Fit = -7.4 + 3.8 x
```

R^2 = 0.996585

Parameter Table:

	Estimate	Standard Error	t-Statistic	P-Value
a	3.8	0.305505	12.4384	0.00111985
b	−7.4	1.58745	−4.66156	0.0186307

```
{a → 3.8, b → -7.4}
```

```
N[Log[e, 10]]
N[Log[e]]
```

```
2.30259
```

```
1.
```